
LIBRAIRIE   IR- REMOTE

La librairie IRRemote crée par Ken-Shirriff est disponible sur GITHUB.
Elle permet de piloter un arduino par n'importe quelle télécommande infrarouge à l'aide d'un 
récepteur du style VS1838B, sortie « signal » connectée sur une entrée numérique de l'arduino.

On peut aussi cloner une télécommande à l'aide d'une led infrarouge, connectée obligatoirement 
sur la broche 3, via une résistance d'environ 100 ohms.
Dans ce cas il faut connaître le protocole de communication (NEC, SONY, RC5, …)
On le trouve facilement avec l'exemple fourni dans la bibliothèque : IrrecvDump.

Importation: #include <Irremote.h>

Réception

Au début du programme:
int broche_reception = 11; // broche utilisée pour la réception
IRrecv reception_ir(broche_reception); // crée une instance nommée reception_ir (par exemple, on 
peut mettre un autre nom)
decode_results decode_ir; // variable de stockage des données reçues

Dans le setup() : reception_ir.enableIRIn(); // démarre la réception

Dans le loop() : if (reception_ir.decode(&decode_ir)) {touche=decode_ir.value;}

decode_ir.value contient le code hexadécimal de la touche enfoncée

reception_ir.resume(); // remise à 0, en attendant le code suivant

Emission

La led émettrice doit être sur la broche 3 avec une résistance d'environ 100 ohms 

Plusieurs protocoles sont dans la bibliothèque : NEC, RC5, SONY, ….

Au début du programme:
IRsend emission_ir; // crée une instance

Dans le setup() : rien

Dans le loop() : emission_ir.sendNEC(0xFB2AD5, 32); 
(Pour le protocole NEC en 32 bits, code envoyé :  0xFB2AD5



Code du récepteur

/* Réception infrarouge de l'émetteur à 2 boutons
la réception d'un code augmente la luminosité de la led (PWM sur 5), l'autre la baisse*/

#include <Irremote.h>

int led = 5;
int puissance = 0; //intensité de la led
int broche_reception = 11; // broche utilisée pour la réception
IRrecv reception_ir(broche_reception); // crée une instance nommée reception_ir
decode_results decode_ir; // variable de stockage des données reçues
unsigned long touche=0; //Contient le code de la touche utilisée

void setup(){
  Serial.begin(9600);
  reception_ir.enableIRIn(); // démarre la réception
  pinMode(led,OUTPUT);
}

void loop(){
  if (reception_ir.decode(&decode_ir)){
    touche=decode_ir.value;
    Serial.println(touche,HEX);
    if (touche==0xFB2AD5){
      puissance+=50;
    }
    if (touche==0xFB6897){
       puissance-=50;
    }
     Serial.println(puissance);
     if (puissance<0){puissance=0;}
     if (puissance>255){puissance=255;}
     Serial.println(puissance);
     analogWrite(led,puissance);
     delay(500);
    reception_ir.resume(); // remise à 0, en attendant le code suivant
  }
}



Code de l'émetteur
/*  Emission infrarouge avec 2 boutons pour envoyer 2 codes, l'un pour baisser, l'autre pour 
augmenter la luminosité
  Diode IR sur 3 obligatoire , avec 100 ohms(biblio)
  une led s'allume au moment de l'envoi*/

#include <Irremote.h>

IRsend emission_ir; // crée une instance
const byte led=7;
const byte inter1=6;
const byte inter2=9;
byte etat_inter1=0;
byte etat_inter2=0;

void setup(){
  pinMode(led,OUTPUT);
  pinMode(inter1,INPUT_PULLUP); // les inters sont en pull_up interne arduino, donc 1 ouvert
  pinMode(inter2,INPUT_PULLUP);
}

void loop(){
    etat_inter1=digitalRead(inter1);
    etat_inter2=digitalRead(inter2);
    if (etat_inter1==0){
      digitalWrite(led,HIGH);
      emission_ir.sendNEC(0xFB2AD5, 32); 
      delay(500);
      digitalWrite(led,LOW);
    }
    if (etat_inter2==0){
      digitalWrite(led,HIGH);
      emission_ir.sendNEC(0xFB6897, 32); 
      delay(500);
      digitalWrite(led,LOW);
    }
}


